
News 
CED visit to Boston 

CED Sales engineer Steven Clifford will be visiting existing and prospective customers in 
the Boston area during the week beginning 27th March. Simon Gray will also be visiting 
Toronto on Wednesday 29th March. If you would like a visit to discuss your current or 
future research applications, or arrange a demonstration of our latest software and 
hardware capabilities, please contact steve@ced.co.uk.   

CED at Experimental Biology 2006 
We will be exhibiting at Experimental Biology 2006 in San Francisco from Sunday 2nd 
to Tuesday 4th of April. If you are attending the meeting, feel free to stop by our stand at 
booth no. 1517. We will also be visiting existing and prospective customers in the 
surrounding area after the meeting so if you are interested in a visit to discuss current or 
future research applications, please contact sales@ced.co.uk. 

 

Spike2 and Signal Training Courses 
Our U.K. training days will be held on the 20th – 21st of March at St. John’s Innovation 
Centre, Cambridge. Core sessions cover sampling, analysis and advanced options such 
as spike sorting and script writing. Workshop sessions are also available for users to 
discuss specific requirements with CED engineers and programmers. There are a few 
places still available for the training days so for further details and a booking form, visit 
the training section of our website. 

 

Latest versions of Spike2 and Signal 
The latest updates for Spike2 and Signal are available from the CED downloads page for 
registered users, or by clicking on the links in the table below. Demonstration versions of 
the latest software are also available. 
 

 Spike2 downloads Signal downloads 
Spike2 version 5.14  Signal version 3.06a 

 Spike2 version 4.24  Signal version 2.16 

 Spike2 demo  Signal demo 

A full list of the new features and changes in the latest software versions is available from 
the website. 

Spike2 

Q. How can I take simple measurements such as mean or curve area from a 
 selected time range of data in Spike2?  

 

A. There are a number of ways to take measurements from data in Spike2, ranging 
 from taking quick manual readings between pairs of cursors to generating 
 automated XY trend plots using Active cursors.  

 The Cursor regions dialog, available from the Cursor menu, is a quick way to 
 take measurements from regions of data in a time or result view.  The dialog 
 calculates and displays values between pairs of cursors based on the 
 measurement type selected in the bottom left of the dialog. Select the calculation 
 to display by clicking on the field and choosing from the drop-down list. The 
 values in the regions dialog are automatically calculated and updated. Results 
 can be copied to the Log view, clipboard or printed by either right-clicking in the 
 dialog or using the associated hotkeys: Ctrl+L for Log, Ctrl+C for Copy or 
 Ctrl+P for Print.  

  

 
  

  

    
  Cursor regions dialog showing the measurement list 

 Script writers can perform all of the available measurement types from the Cursor 
 regions  dialog using the ChanMeasure()function. Further information about  
 this command can be found in the script section of the on-line help. 

 Hint: Spike2 also has an interactive measurement function for fast reading of 
 time range and amplitude data from a channel. Simply hold Alt and click and 
 drag the mouse pointer on a data channel to access this.  

Signal 

 

Q. I am starting a long term potentiation (LTP) study and would like to record 
 responses during a period of control stimuli, until the response stabilizes, 
 and then apply a ‘one shot’ pulse train before reverting back to the control 
 stimulus, as below.  

  
State 1 control stimulus 

 

  
State 2 train stimulus 

 I have tried to set-up the stimuli using the dynamic outputs mode of multiple 
 states using State 1 to setup and prepare the recording using the control stimulus 
 and State 2 for the pulse train. I need manual control over the Basic and control 
 states but would like the train output to automatically revert back to State 1 after 
 a single cycle. Is there an easy way to do this? 

 

A. Dynamic outputs can use DAC or digital outputs to control an experiment where 
 each State generates a different set of pulse outputs during sampling. The States 
 toolbar shows the current condition and gives manual control of selection using 
 the toolbar buttons or the drop-down list. The Basic 0 state can be used as a 
 default ‘idle’ state during the recording, and can be selected using the B0 or Idle 
 button. It is also possible to set-up a sequence to determine the order in which to 
 apply the various outputs during a recording. 

             
States toolbar 

 The ordering for dynamic outputs is set using the drop-down list in the States tab 
 of the sampling configuration. States can be set to run in a numeric or random 
 order where the user simply specifies the number of repeats. They can also be 
 set to follow a protocol where the user decides the order and number of repeats 
 for the outputs. A protocol consists of a list of up to ten steps, where each step 
 specifies a state number, a repeat count and the next step to use. For this 
 example we want to be able to order the outputs so that whenever State 2 is run, 
 it returns back to State 1 after a single cycle.  

 
 

  

 

  

 

 

 

 

States setup in the sampling configuration with Protocol sequencing 

 Protocol execution always starts at step 1 when the Cycle toolbar button is 
 pressed. The sequence above is set to run State 2 once, and then go to step 2. 
 This step sets the State back to 1 and then stops the protocol (by setting the 
 next field to 0). This will always return the output to our control after a train 
 stimulus, while still allowing manual selection of the Basic and control States in 
 the experiment.  

Scripts: Spike2          (double click to open) 
Q. I would like to be able to create short ‘tone’ waveforms to output during 
 recording. Is there a way to do this in Spike2, either interactively or via a script? 

A. The attached script, TonePip.s2s can be used to create tone pips and tone 
 bursts for playback from a DAC output when sampling. The script creates a 
 toolbar from which the user can generate single or sequences of tone pips of a 
 defined  frequency and duration. Up to ten different tones can be created and are 
 loaded into the Play waveform area of the current sampling configuration for use 
 when sampling, or saved to a new data file for later use. This script requires the 
 latest Spike2 version 5, and supports the 1401plus, micro1401, Micro1401 mk II 
 and Power1401 for output. 

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know.  



'$Tone_Pip|Create tone pips for playback during sampling via a dac output
'=============================================================================================================
'CAMBRIDGE ELECTRONIC DESIGN LIMITED, THE SCIENCE PARK, MILTON RD., CAMBRIDGE CB0 0FE, UK
'=============================================================================================================

'  Copyright © Cambridge Electronic Design,  Feb. 2006.

'SYSTEM REQUIREMENTS:
'   Hardware: All 1401 models except Standard 1401.
'   Software: Spike2 v5.13 or higher
'   Script: TonePip.s2s

' HEALTH WARNING
'   This script is a 'work in progress' and is offered without guarantees.
'   You must test it to see whether it is suitable for your application

'OVERVIEW:
'   This script creates tone pips and tone bursts, Sine waves at a chosen frequency are amplitude modulated with an envelope that consists of a
'   linear rising phase, a plateau and a linear falling phase. You can choose the plateau amplitude and the duration of rise, plateau and fall phases
'   via a dialog.You can display the tone pip in a time view or add it to the PlayWave area of the current sampling configuration so that you can play the tone
'   during sampling, via one of the DAC outputs. When creating PlayWaves, you have the option to set an interval following the tone pip before it can play again and the number
'   of repetitions of the pip for each button press. This feature allows you to create, for example, a model cricket chirp consisting of 4 x  20ms pips @5kHz
'   separated by 20ms intervals.
'
'   The fidelity of sound playback will depend on the model of 1401 interface that you are using.
'   The Power 1401 has 16 bit Dacs, that is the output voltage range is subdivided into 65536 steps.
'   Assuming a DAC range of +-5V, this translates to minimum step changes in amplitude of 0.16mV.
'   Other models of 1401 have 12 bit dacs giving a minimum increment of approx. 2.5mV.

'   The maximal time resolution of the playback (equivalent to maximum sampling rate for recorded data) is also dependent on the model of 1401.
'   This script limits the minimum step for the Power 1401 and Micro Mk2 to 10us (equivqlent to 100kHz sampling rate). For other models it is set to 50us (20kHz).
'   The maximum tone pip duration will depend on the amount of 1401 memory installed less 250kB which Spike2 reserves for recording.
'   See the Help section on Arbitrary waveform output for further information.

'USER GUIDE
'   Run the script via the RunScript/Load and Run option on the Spike2 Script drop-down menu. Alternatively, you could create a button on the 
'   Script bar. Consult the online help for guidance on how to do this
'   The script toolbar has 3 buttons:

'   QUIT: simply closes the script and restores the display to its former state.

'   CLEAR ALL PLAYWAVES: also works as advertised. -It deletes all PlayWaves from the current sampling configuration. You can also delete
'   play waves via the controls on the sampling configuration control panel.

'   MAKE PIP: This button opens a dialog for you to create a new tone pip and display it in a time view or add it to the sampling configuration for 
'   online playback.The dialog has the following sections:

'   GENERAL: Here you specify the type of 1401 and the destination of the pip, - either a time view or the Play wave area.
'   If play wave is selected, you need to specify the dac output to use, the label for the button on the playwave bar and a character on the keyboard
'   that will trigger the pip. Ten keyboard characters are available. These are taken from the bottom row of characters on a QWERTY style keyboard. 

'   TONE PARAMETERS: Here, you select the time resolution of the playback and the carrier frequency and amplitude (zero to peak) of the sine wave.

'   ENVELOPE PARAMETERS: Here you select the duration of each component of the envelope.

'   REPETITION: If you require a sequence of pips , enter the interval between pips and number of repetitions here.
'   If you select 0 repetitions, then a continuous train of pips will play until explicitly stopped by clicking on the 'No output' button.
'   If you require a single pip, then set the inter-pip interval to zero.
'   If you check the External trigger box, the system will be "armed" when you press the PlayWave button or its short-cut key but the wave will
'   only be played when the 1401 receives a trigger pulse at the trigger input on the front panel (E3 input on 1401plus). This allows you to generate
'   precisely timed single pulse.However, note that you still will only get one pip per button press.

'   When you click on OK, the tone pip will be added to the sampling configuration or displayed in a time view depending on what you selected.
'   You will be prompted to save the updated sampling configuration so that the play waves that you added will be available in future.
'   A warning message will appear if the chosen time resolution is not compatible with your hardware and the script will use the nearest (higher)
'   valid resolution.

'   You are free to add several different tone pips to the play wave bar. Simply choose different button labels and short-cut keys for each
'   The script will prompt you to save the updated sampling configuration so that the new Play waves will be available next time you use it.
'   When you have finished creating pips, press QUIT to close the script.

'   PLAYING WAVEFORMS: 
'    To play the tone pip or sequence of tone pips once, click on the appropriate short-cut key or PlayWave button while sampling.
'   If you want to play the waveform that you have created (or any other stored waveform) repetitively, you can use the Wave_Play script.
'   For example you could use the Wave_Play script to output a model cricket chirp at regular intervals of, say, 500ms.

var tflash;'                                         used by Newsflash function
var gFloat%[20];'                                used by Hide/RestoreApps()
var amp:=1.0;'                                     default amplitude of pip
var period,f:=4800,stimdur;'             sine period, frequency and pip duration
var rampup:=2,rampdwn:=2,platodur:=21,intvl:=25;' pip duration parameters
var tr%:=10,tres; '                            time resolution in micoseconds and seconds
var btnlbl$:="pip";'                           label for the playwave button
var dac%:=0;'                                       play from this dac
var ok%;
var symmchk%;'                                     check for asymmetrical pips
var stimkey$:="z";'                            key that plays the waveform
var model$[4],modelndx%;'                 model of 1401 in use
var dest$[2],destndx%:=1;'               destination for the wave
var nreps%:=1;'                                  number of times to play the wave
var trigchk%;

model$[0]:="Power";model$[1]:="Micro mk.II";
model$[2]:="Micro";model$[3]:="1401plus";
dest$[0]:="Time view";dest$[1]:="Play Wave area";

HideApps();'hide unwanted toolbars etc.
ToolbarSet(1,"&Quit",Quit%);
ToolbarSet(2,"&Make Pip",MakePip%);
ToolbarSet(3,"&Clear All play-waves",ClearPips%);
Toolbar("",511);
halt

func Quit%()'quit the script
RestoreApps();
return 0;
end;

func MakePip%()
var msg$,tst1,tst2,tst3,flag%:=0,pw$,q%;
var keylist$[10],keyndx%;
keylist$[0]:="z";keylist$[1]:="x";keylist$[2]:="c";keylist$[3]:="v";keylist$[4]:="b";'bottom row of qwerty keboard +jkl.
keylist$[5]:="n";keylist$[6]:="m";keylist$[7]:="j";keylist$[8]:="k";keylist$[9]:="l";

pw$:=PlayWaveInfo$();' list of current PlayWave keys

DlgCreate("Create a tone pip:",0,10,88);
DlgAllow(511,0,Ch%);
DlgGroup("General:",1,1,44,6);
DlgList(1,"Type of 1401:",model$[],4,24,2);
DlgList(2,"Destination:",dest$[],2,24,3);
DlgInteger(3,"Play via DAC:",0,3,30,4,1);
DlgString(4,"Label for Play Wave button:",8," a-zA-Z0-9",32,5);
DlgText("Hotkey to play the tone pip",0,6);
DlgList(5,5,keylist$[],10,37,6);
DlgGroup("Tone parameters:",1,7.5,44,4);
DlgText("Time resolution (us)",0,8.5);
DlgInteger(6,12,10,1000,30,8.5,1);
DlgText("Carrier frequency (kHz)",0,9.5);
DlgReal(7,12,1.0,10000.0,30.0,9.5,100.0);
DlgText("Plateau amplitude (V):",0,10.5);
DlgReal(8,12,0.0,5.0,30,10.5,0.1);

DlgGroup("Envelope parameters",48,1,39,6);
DlgCheck(9,"Check for asymmetric pips",50,2);
DlgText("Linear rise duration (ms):",50,3);
DlgReal(10,8,0.1,10000.0,0,3,1.0);
DlgText("Linear fall duration (ms):",50,4);
DlgReal(11,8,0.1,10000.0,0,4,1.0);
DlgText("Plateau duration (ms):",50,5);
DlgReal(12,8,0.1,10000.0,0,5,1.0);

DlgGroup("Repetition:",48,7.5,39,4);
DlgText("Interval between pips (ms);",50,8.5);
DlgReal(13,8,0.0,10000.0,0,8.5,1.0);
DlgText("Number of cycles (0:= endless):",50,9.5);
DlgInteger(14,8,0,10000,0,9.5,1);
DlgCheck(15,"Check for External triggering.",50,10.5);
ok%:=DlgShow(modelndx%,destndx%,dac%,btnlbl$,keyndx%,tr%,f,amp,symmchk%,rampup,rampdwn,platodur,intvl,nreps%,trigchk%);
if not ok% then return 1 endif;
Yield();'give time for dialog to close

stimkey$:=keylist$[keyndx%];
if InStr(pw$,stimkey$) <> 0 then
    q%:=Query(Print$("Key %s already exists. Do you want to overwrite it?",stimkey$));
    if q% = 0 then return 1
    endif;
endif;

if modelndx% > 1 then 'plus or Micro Mk1
    repeat
        tst1:=tr% mod 4;' check that chosen step meets resolution limitations  
        tst2:=tr% mod 6;
        tst3:=tr% mod 10;
        if tst1 <> 0 and tst2  <> 0 and tst3 <> 0  then 'step is incompatible with sampling
            flag%:=1;'
            tr%+=1;
        endif;
    until tst1 = 0 or tst2 = 0 or tst3 = 0;
    if flag% then
        Message("Closest match to chosen time resolution: %d us will be used.",tr%);
    endif;
endif;
tres:=tr%;
Tres/=1000000.0;'convert time resolution us to s.
period:=1.0/f;
rampdwn/=1000.0;'convert pip durations from ms to s
rampup/=1000.0;
platodur/=1000.0;
intvl/=1000.0;
stimdur:=rampup+platodur+rampdwn;
MakeTone(period,destndx%);
if destndx% > 0 then '  if  saving to play wave area...
    ok%:=FileSaveAs("",6,1,"Save configuration as:");'save updated sampling configuration
    if ok% = 0 then
        msg$:="Configuration saved successfully.";
    else
        msg$:="Configuration not saved.";
    endif;    
    NewsFlash(msg$,2.0,0,0,0);
endif;
rampdwn*=1000.0;'convert pip durations from s to ms as defaults for next time the dialog is called
rampup*=1000.0;
platodur*=1000.0;
intvl*=1000.0;
return 1;
end;

func Ch%(item%);'                                                        change function for the MakePip() dialog (enabling and disabling items etc.)
var en%;

if item% = 1 or item% = 4 then
	if DlgValue(1) > 1 and DlgValue(4) < 50 then
		NewsFlash("Minimum time resolution for chosen harware is 50us",2.0,0,30,0);
		DlgValue(6,50);
	endif;
endif;
if item% = 0 or item% = 2 then 'enable playwave set up items if required
    if DlgValue(2) = 0 then
        DlgEnable(0,3,4,5,13,14,15);
    else
        DlgEnable(1,3,4,5,13,14,15);
    endif;
endif; 
en%:=DlgValue(9);
DlgEnable(en%,11);'en/dis-able fall duration depending on state of check box
if en% = 0 then
    DlgValue(11,DlgValue(10));'set fall =rise
endif;    
return 1;
end;

func ClearPips%()
PlayWaveDelete(PlayWaveInfo$());'delete all play waves in current sampling configuration
View(App(5)).WindowVisible(0);'hide playwave bar
return 1;
end

Proc MakeTone(per,dest%);       'tone burst, optional asymmetrical rise and fall. per is period of sine wave.
var pi,stimarrsz%,ptspercycle;
var angle,i,rate;
var risesz,fallsz,rmparrsz%,intvlsz;
'	var pwclear[stimmaxdur/tres];'array of zeros to clear the playwave area

pi:=4.0*ATan(1.0);
'calculate arraysizes using requested temporal resolution
risesz:=Trunc(rampup/tres);
fallsz:=Trunc(rampdwn/tres);
rmparrsz%:=Max(risesz,fallsz);'set size of array to hold the longer ramp
stimarrsz%:=Trunc(stimdur/tres);'array size
intvlsz:=Trunc(intvl/tres);
var ramp[rmparrsz%],stimarr[stimarrsz%+intvlsz];'array large enough for pip and  following interval

'generate sinewave by array arithmetic
ptspercycle:=per/tres;'	                                    sample points per cycle
angle:=2*pi/ptspercycle;'	                            angular shift between sample points in radians
ArrConst(stimarr[:stimarrsz%],angle);'             fill array with angle
ArrIntgl(stimarr[:stimarrsz%]);'	                     convert array vals to angle*i%
Sin(stimarr[:stimarrsz%]);'	                            take sine of angles
ArrMul(stimarr[:stimarrsz%],amp);'	              multiply by amplitude

var inc;
inc:=1.0/risesz;
ramp[0]:=0.0;
ArrConst(ramp[1:risesz-2],inc);
ArrIntgl(ramp[]);'	                       linear rising ramp
ArrMul(stimarr[],ramp[:risesz]);'multiply wvform by rising ramp
inc:=1.0/fallsz;
ramp[0]:=0.0;
ArrConst(ramp[:fallsz],inc);	                        
ArrIntgl(ramp[:fallsz]);'	                        linear falling ramp
ArrSubR(ramp[:fallsz],1.0);
ArrMul(stimarr[stimarrsz%-fallsz:],ramp[:fallsz]);'

if dest% = 0 then'write wave to a time view
    var ovwh%,bsz,arrsz%,tofs:=0.005;
    ovwh%:=FileNew(7,0,1,2,stimdur+intvl);
    WindowTitle$("Pip");
    ChanNew(1,1,0,tres);
    ChanShow(1);
    Window(50,0,100,100);'right half of screen
    bsz:=BinSize(1);
    arrsz%:=(stimdur+intvl)/bsz;
    var arr[arrsz%];
    ArrConst(arr[],0.0);
    ChanWriteWave(1,arr[],0.0);'create a baseline
    ChanWriteWave(1,stimarr[],tofs);'add copy of the pip 
    XRange(0,tofs+stimdur+intvl+0.01);
    CursorSet(5,tofs,tofs+rampup,tofs+rampup+platodur,tofs+rampup+platodur+rampdwn,tofs+rampup+platodur+rampdwn+intvl);
    for i:= 1 to 5 do
        CursorLabelPos(i,2);
    next;    
    Optimise(1);
    DrawMode(1,13);' cubic spline mode
    WindowVisible(1);
else 
	ok%:=PlayWaveAdd(stimkey$,btnlbl$,dac%,1/tres,stimarr[]);'  
	PlayWaveEnable(stimkey$,1);
    PlayWaveCycles(stimkey$,nreps%);' number of cycles to do
    PlayWaveTrigger(stimkey$,trigchk%);    
	View(App(5)).WindowVisible(1);'show playwave bar
endif;
return;
end;

proc NewsFlash(msg$,Tshow,x,y,beep%);'                    floating message box at coordinates with optional audible alert
var dummy;
if Tshow < 1.5 then Tshow:=1.5; endif;'set minimum display time
tflash:=Tshow;'tflash is a global variable
if beep% then
	Sound("S*",1);'beep
endif;
Seconds(0);'reset the clock
DlgCreate("Info:",x,y);
DlgLabel(1,msg$);
DlgAllow(1023,NFIdle%);'Idle routine does the timing
DlgButton(0,"");'hide cancel button
DlgShow(dummy);
return;
end;

func NFidle%();'                                                          Newsflash idle routine
var ret%:=1;
if Seconds() > tflash then' times up 
	ret%:= 0;'             so close message box
endif;
return ret%;
end;

Proc HideApps();'                                                        Hide unnecessary toolbars etc. while script is running.
var i%;
View(App(3)).WindowVisible(1);'normalise script view before hiding (may reappear if maximised)
View(App(5)).WindowVisible(1);'show playwave bar
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
return;
end;

Proc RestoreApps();'                                                    Restore toolbars etc. to former glory
var i%;
for i%:=1 to gFloat%[0] do
    docase
   	case i%=3 then
		View(App(3)).Windowvisible(0);'keep script invisible
   	case i%=5 then 
		View(App(5)).Windowvisible(1);' keep playwave bar visible        
	else
		View(App(i%)); 'restore other items to previous state
		WindowVisible(gFloat%[i%]);
	endcase;
next;
return;
end;

Scripts: Signal          (double click to open)  
Q. I have a number of files recorded from a series of experiments and I would like to 
 amalgamate these multiple files into a single file for analysis, so I don’t have to 
 analyse each file in turn. Is there a script available for this? Because I use 
 multiple states, it is important that the state information is also copied for each 
 frame when the files are joined together. 

A. The attached script, AppendFile.sgs can be used to ‘join’ multiple data files 
 together. The user is prompted to open a target data file, and then a source data 
 file to append to the target. The script expects the files to contain the same 
 number of channels, and be sampled at the same rate. Each frame of the source 
 file, along with associated state information, is then appended to  the target file. 
 These appended frames are only held in memory, you need to save the target 
 file to make them permanent.  

Did you know…? 
There is a feature in the Y-axis dialog that lets you set the displayed range to be 
symmetrical around 0. Simply type in the positive value to use in the Top field and click 

the  button to copy this value inverted to the Bottom field. You can also copy the 
Bottom value to the Top field using the corresponding button.  

                 

Y-range dialog 

Recent questions  
Q. Are there keyboard shortcuts in Spike2 for optimising the data in all channels in a 
 file? This would be much easier than opening the Y-axis dialog and setting all 
 channels to optimise. 

A. There are many functions in both Spike2 and Signal that have associated 
 ‘hotkey’ keyboard shortcuts. For this example we can make use of two such 
 shortcuts in Spike2. Pressing Ctrl+A will select all visible channels in a data, 
 result or XY-view. Pressing Ctrl+Q will then optimise the selected channels. 

 A full list of keyboard shortcuts can be found in the on-line help for both Spike2 
 and Signal. 

 

User Group  

If you have any comments about the newsletter format and content, or wish to un-
subscribe from the mailing list for this newsletter, please notify sales@ced.co.uk. 

Back to top 
Right-click the script icon 
and save to disk. 

If you have any problems 
opening the embedded 
scripts in this newsletter 
please let us know. 

'$AppendFiles|Select a signal data file to append to a master file
'The script copies waveform and marker data from one file to new
'frames appended to the end of a master file.

var file1%;		'handle for master file
var file2%;		'handle for file to append
var noFrames1%;	'frame count from master file
var noFrames2%;	'frame count from file to append	
var c%;			'counter
var state%;		'frame state 
var gFloat%[20];              'global for floating window states

HideAll();
file1%:= FileOpen("",0,0,"Select master file");		'Prompt user to open master file
noFrames1%:= FrameCount();		'Get frame count
file2%:= FileOpen("",0,0,"Select file to append");	'Prompt user to open file to append
noFrames2%:= FrameCount();		'Get frame count

View(file1%);	'Make sure we are looking at master file

for c%:=1 to noFrames2% do		'For each frame from file 2
	AppendFrame();		'Append a new frame to master file
	View(file1%).Frame(noFrames1%+c%);		'Select the newly appended frame
	View(file2%);		'Get file2
	Frame(c%);			'Get the corresponding frame from file 2
	state%:= FrameState(c%);	'Get the frame state
	ExportChanFormat(0,1,1,1);	'Set the waveform data to copy
	ExportChanFormat(1,0,0,0);	'Set the marker data to copy
	ExportTextFormat(5,5,0,"/","");		'Get data values for copying
	ExportChanList(-1);			'Get all channels
	ExportFrameList(-2);		'Get the current frame
	ExportTimeRange(0.0,Maxtime());	'Get the data from the whole frame
	EditCopy(15);	'Copy as text
	FrontView(file1%);	'bring the master file to the front
	EditPaste();	'paste in the frame data
	FrameState(-2,state%);	'copy the frame state
	Draw();
	Optimise(-1);
next;

RestoreAll();

proc HideAll()
var i%;
gFloat%[0] := App(-2);        'number of windows
for i% := 1 to gFloat%[0] do  'hide all windows and save state
   gFloat%[i%] := View(App(i%)).WindowVisible(0);
   next;
end

proc RestoreAll()
var i%;
for i% := 1 to gFloat%[0] do  'restore hidden windows
   View(App(i%)).WindowVisible(gFloat%[i%]);
   next;
end

